5. ADA GYAKORLAT

(GENERIC, PARAMÉTEREZÉS, PÉLDÁNYOSíTÁS)

(verem, szumma, rendezés)

Példa - HF  

http://en.wikibooks.org
Valósítsd meg a Sor típust korlátozott átlátszatlan típusként egy

  könyvtári egységként elkészített csomagban! A reprezentációhoz használj

  diszkriminánsos rekordot és megszorítás nélküli indexhatárú tömböt!

sor: peldak / sor / 

-------------------------------------------------------------------------------------------------------------------------------------

Generic-ek

- Sablon:  egy minta

- alapján végrehajtható egyedek hozhatók létre (példányok)

-a példányosítás mindig fordítási időben zajlik

- sablon definíciója: - formális paraméterek megadása

-sablon specifikációja

-sablon törzse

-példány megádasa (formális paramétek aktuálissal való helyettesítése)

 - több példány is létrehozható

- minden példány önálló fordítási egység (alprogram sablon - meghívhító alprogram ;

csomag sablon -  csomag)

1. Példa: alprogram sablon


generic
type Szám is private; -- paraméter
with function Muvelet(A, B: Szám) return Szám; -- paraméter
function Iterált(A: Szám; N: Positive) return Szám; -- deklaráció

function Iterált(A: Szám; N: Positive) return Szám is -- törzs
X: Szám := A;
begin
for I in 2..N loop
X := Művelet(X,A);
end loop;
return X;
end Iterált;
2. Példa: csomag sablon

generic 

type Elem is private; -- paraméter
------------- innen ugyanaz, mint a package, de a paramétereket használhatom

package VEREM is     -- verem tipus vektorban abrazolva

        type Verem (Max: Positive) is limited private;

        procedure Push  ( v: in out Verem; e: in Elem );

        function  Top   ( v: Verem ) return Elem;

        ...

private

      type Tömb is array( Integer range <> ) of Elem;

      type Verem(Max: Positive ) is record

                                        Tartalom: Tömb(1..Max);

                                        Mutato: Natural := 0;

                                    end record;

end VEREM;

package body VEREM is

        procedure Push ( v: in out Verem; e: in Elem ) is

        begin

                v.mutato := v.mutato+1;

                v.tartalom(v.mutato) := e;

        end Push;

        function  Top  ( v: Verem ) return Elem is

        begin return v.tartalom(v.mutato); end Top;

end VEREM;

Paraméterezés

Mi lehet generic paramétere:

•Objektum:

E: in Integer:=42; -- másolat szerinti átadás

F: in out Integer; -- hivatkozás szerinti átadás 

•Tetszőleges tipus:

type T1 is private; - általános típust jelöl 

type T2 is limited private;

• Diszkriminánsos típus:

type T(D1: T1; D2:T2) is limited private;

• Származtatott tipus:

type TN is new T;

• Diszkrét tipusok:

type T is (<>);

•Előjeles egész tipusok:

type T is range <>;

• Lebegőpontos tipusok:

type T is digits <>;

• Fixpontos tipusok:

type T is delta <>;

• Tömb tipusok:

type T1 is array(Index) of Elem;

type T2 is array(Index range <>, Index) of Elem;
• Eljárás paraméter:

with procedure P(X: in out Integer);

• Függvény paraméter:

with function F(X: T1) return T2;

with function ToString(E: T) return String is T’Image;

with function "<" (A,B: T) return Boolean is <>;

•Csomag:

generic

         with package Aritmetika is new Mátrix_Aritmetika(<>);  -- egy példány!

package Összetett_Mátrix_Mûveletek is       ----- ez a sablon neve

         use Aritmetika; ...
Példányosítás

-A "with" "new" és "use" kulcsszavak használatával 

- a use-t csak a példányosítás után lehet használni

- formális és aktuális paraméterek megfeleltetése

 3. Példa:

   A 2. példa verem_vg nevű sablonjának példányosítása egy DEMO nevű eljárásban

with VEREM; 

procedure Demo is

        package INT_VEREM is new VEREM(INTEGER); use INT_VEREM;

        v: Verem(250);   



begin

   .... 

Példa1.   vermek -- Generic package, Private paraméter 

Valósítsd meg a verem típust sablon csomagban a szokásos műveletekkel! 

A sablon paramétere legyen:

- az elemek típusa, mint átlátszatlan (private)  típus, 

Maga verem legyen korlátozott átlátszatlan, amit diszkriminánsos rekord segítségével valósíts meg (a verem hossza a diszkrimináns). A rekord mezői pedig legyenek egy tömb, ami az elemeket tartalmazza, és egy mutató (veremtető), ami a verem tetejére mutat. Kezdetben a

verem legyen üres, minden újonnan deklarált változó esetén. 

A szokásos (Push, Pop, Top) műveletek mellett legyen még egy-egy függvényünk a verem ürességének illetve telítettségének vizsgálatára, valamint a verem aktuális méretének kiszámolására.!

vermek.ads

generic

      type Elem is private;

package Vermek is

      type Verem (Max: Positive) is limited private;

      procedure Push( V: in out Verem; E: in Elem );

      procedure Pop( V: in out Verem; E: out Elem );

      function Top( V: Verem ) return Elem;

      function Is_Empty( V: Verem ) return Boolean;

      function Is_Full( V: Verem ) return Boolean;

      function Size( V: Verem ) return Natural;

private

      type Tömb is array( Integer range <> ) of Elem;

      type Verem(Max: Positive ) is record

                                        Adatok: Tömb(1..Max);

                                        Veremtető: Natural := 0;

                                    end record;

end Vermek;

-------------------------------------------------------------------------------------------------------------

vermek.adb

package body Vermek is

      procedure Push( V: in out Verem; E: in Elem ) is

      begin

          V.Veremtető := V.Veremtető + 1;

          V.Adatok(V.Veremtető) := E;

      end;

      procedure Pop( V: in out Verem; E: out Elem ) is

      begin

          E := Top(V);

          V.Veremtető := V.Veremtető - 1;

      end;

      function Top( V: Verem ) return Elem is

      begin

          return V.Adatok(V.Veremtető);

      end;

      function Is_Empty( V: Verem ) return Boolean is

      begin

          return V.Veremtető = 0;

      end;

      function Is_Full( V: Verem ) return Boolean is

      begin

          return V.Veremtető >= V.Max;

      end;

      function Size( V: Verem ) return Natural is

      begin

          return V.Veremtető;

      end;

end Vermek;

-------------------------------------------------------------------------------------------------------------

integer_vermek.ads

Hozzunk létre külön fordítási egységben  egy verem csomag példányt integer típusú elemek kezelésére. 

with Vermek;

package Integer_Vermek is new Vermek(Integer);

-------------------------------------------------------------------------------------------------------------

megfordit.adb

Hozzunk létre a fenti Integer_Vermek csomag példány és egy itt létrehozott Int_Vermek csomag példány segítségével egy-egy vermet. A parancssorban megadott argumentumokat tegyük bele az egyik verembe, majd egyenként átpakolva őket a másik verembe, és utána kiírva az utóbbi verem tartalmát az elemeket az eredeti sorrendben kapjuk vissza. 

with Vermek, Integer_Vermek, Ada.Command_Line, Ada.Integer_Text_IO;

use Integer_Vermek; use Ada.Integer_Text_IO;

procedure Megfordit is

    package Int_Vermek is new Vermek(Integer); 

    use Int_Vermek;

    N: Integer;

    V:Int_Vermek.Verem(Ada.Command_Line.Argument_Count);

   W:Integer_Vermek.Verem(Ada.Command_Line.Argument_Count);

begin

    for I in 1..Ada.Command_Line.Argument_Count loop

       N := Integer'Value(Ada.Command_Line.Argument(I));

       Push( V, N );

    end loop;

    while not Is_Empty(V) loop 

        Pop( V, N );

        Push( W, N );

    end loop;

    while not Is_Empty(W) loop 

        Pop( W, N );

        Ada.Integer_Text_IO.Put(N);

   end loop;

   get(N);

end Megfordit;

-------------------------------------------------------------------------------------------------------------

Példa2 szumma --- Generic alprogram -- függvény, egyéb típusú paraméterek 

Írj sablon függvényt Fold névvel, amely egy vektor, elemeinek Op-át számolja ki (Pl."+"-át) .

A sablon paramétere legyen:

- a vektorelemek típusa,

- a vektorindexek típusa (diszkrét),

- a vektor típusa,

- az  egység elem (start)

- egy Op  nevű művelet 

A Fold nevű  függvény egyetlen paramétere a vektor legyen, visszatérési értéke

pedig egy elem típusú szám legyen, a vektor elemeire alkalmazva az Op műveletet.

fold.ads

generic

    type Elem is private;

    type Index is (<>);

    type Tömb is array ( Index range <> ) of Elem;

    --type Tömb is array ( Index ) of Elem;

    with function Op( A, B: Elem ) return Elem;

    Start: in Elem;

function Fold ( T: Tömb ) return Elem;
-------------------------------------------------------------------------------------------------------------

fold.adb

function Fold ( T: Tömb ) return Elem is

    S: Elem := Start;

begin

    for I in T'Range loop


    S := Op( S, T(I) );

    end loop;

    return S;

end Fold;

-------------------------------------------------------------------------------------------------------------

szummazas.adb

Írj programot a sablon felhasználásával, amely kiszámítja az (3,2,5,7,1) egészekből álló  (konstans) vektor elemeinek  összegét és szorzatát.

with Fold, Ada.Text_IO;

use Ada.Text_IO;

procedure Szummazas is

    type Index is new Integer;

    type Elem is new Integer;

    type Tömb is array (Index range <>) of Elem;

    function Sum is new Fold(Elem, Index, Tömb, "+", 0);

    function Prod is new Fold(Elem, Index, Tömb, "*", 1);

begin

    Put_Line( Elem'Image( Sum((3,2,5,7,1)) ) );

    Put_Line( Elem'Image( Prod((3,2,5,7,1)) ) );

end Szummazas;

-------------------------------------------------------------------------------------------------------------

Példa3 cserel -- generic alprogram

Írjunk sablont két adott "T" típusú elem felcserélésére. A sablon paramétere a T típus legyen.

cserel.ads

generic

    type T is private;

procedure Cserel ( A, B: in out T );

cserel.adb

procedure Cserel ( A, B: in out T ) is

    Tmp: T := A;

begin

    A := B;

    B := Tmp;

end Cserel;

--------------------------------------------------------------------------

Példa4 maxhely -- generic alprogram

Írjunk sablont egy vektor maximális helyének megkereséséhez. A sablon paraméterei legyenek: 

- a vektorelemek típusa,

- a vektorindexek típusa (diszkrét),

- a vektor típusa,

- egy "<" függvény, amely az összehasonlítást végzi

A Max_Hely függvény bemenő paramétere legyen a tömb, és adja vissza a maximális elem helyének indexét. 

max_hely.ads

generic

    type Elem is limited private;

    type Index is (<>);

    type Tömb is array (Index range <>) of Elem;

    with function "<" ( A, B: Elem ) return Boolean is <>;

function Max_Hely ( T: Tömb ) return Index;

max_hely.adb

function Max_Hely ( T: Tömb ) return Index is

    Mh: Index := T'First;

begin

    for I in T'Range loop

        if T(Mh) < T(I) then

            Mh := I;

        end if;

    end loop;

    return Mh;

end Max_Hely;

--------------------------------------------------------------------------

Példa5 Rendez -- generic alprogram

Írjunk sablont egy vektor elemeinek rendezésére, úgy, hogy az előző Max_Hely és Csere sablonokat használjuk. A sablon paraméterei legyenek: 

- a vektorelemek típusa,

- a vektorindexek típusa (diszkrét),

- a vektor típusa,

- egy "<" függvény, amely az összehasonlítást végzi

A Rendez függvény be és kimenő paramétere legyen a rendezendő ill. a rendezett tömb.

rendez.ads

generic

    type Elem is private;

    type Index is (<>);

    type Tömb is array (Index range <>) of Elem;

    with function "<" ( A, B: Elem ) return Boolean is <>;

procedure Rendez ( T: in out Tömb );

rendez.adb

with Max_Hely, Cserel;

procedure Rendez ( T: in out Tömb ) is

    procedure Cserél_Elem is new Cserel(Elem);

    function Max_Hely_Tömb is new Max_Hely(Elem,Index,Tömb);

    Mh: Index;

begin

    for I in reverse T'Range loop

        Mh := Max_Hely_Tömb( T(T'First..I) );

        Cserél_Elem( T(I), T(Mh) );

    end loop;

end Rendez;

--------------------------------------------------------------------------

Példa5 Rendezés -- FŐprogram

Írjunk demo programot, amely a fenti csomagot használva alkalmas a (3.0,6.2,1.7,5.2,3.9) tömb növekvő illetve csökkenő sorrendbe történő rendezésére.
with Ada.Text_IO, Rendez;

use Ada.Text_IO;

procedure Rendezes is    

    type Tömb is array (Character range <>) of Float;

    procedure Rendez_Növ is new Rendez(Float,Character,Tömb);

    procedure Rendez_Csökk is new Rendez(Float,Character,Tömb,">");

    T: Tömb := (3.0,6.2,1.7,5.2,3.9);

begin

    Rendez_Csökk(T);

    for I in T'Range loop

       Put_Line( Float'Image( T(I) ) );

    end loop;

end Rendezes;

PéldaHF

Házi feladat

Írjunk egy eljárás sablont a TEXT_IO programcsomag PUT_LINE eljárását használva különböző típusú értékekek kiíratására. A sablonnak két paramétere legyen:

- T típus a kiírandó elem típusa (korlátozott private)

- toString nevű fgv., melynek T a bemenő paraméterének típusa és egy stringet ad vissza.

.ads





.adb








